
Declaration

(Incomplete) Declaration Syntax

〈adt name〉: DATATYPE

BEGIN[
〈constructor i〉

[
(
[
〈accessor i,j〉: 〈type〉

]j=1··n
(,)

)
]
: 〈recognizer i〉

]i=1··c

END 〈adt name〉
c, n ∈ N

Given as a collection of constructors

with associated recognizers and accessors

No name repetition allowed

adt name 6= constructori, adt name 6= accessorij , adt name 6= recognizeri
i 6= k ⇒ constructori 6= constructork

i 6= k ⇒ recognizeri 6= recognizerk
i 6= k ⇒ constructori 6= recognizerk
j 6= k ⇒ accessorij 6= accessorik

Declaration
Example

(Incomplete) Declaration Syntax

〈adt name〉: DATATYPE

BEGIN[
〈constructor i〉

[
(
[
〈accessor i,j〉: 〈type〉

]j=1··n
(,)

)
]
: 〈recognizer i〉

]i=1··c

END 〈adt name〉
c, n ∈ N

Example: Phonebook
Phonebook: DATATYPE

BEGIN

empty_phonebook: empty?

add_phone(

name:string,

phone:string,

pb:Phonebook):

add_phone?

END Phonebook

constructors:

empty_phonebook, add_phone

recognizers:

empty?, add_phone?

accessors:

name, phone, pb

CASES Expressions

PVS provides support for a simple form of pattern-matching

CASES x OF[
〈constructor i〉

[
(
[
〈variablei,j〉

]j=1··ni

(,)
)
]
: 〈expression〉

]i∈[1··c]
[
ELSE 〈expression〉

]
ENDCASES

ELSE can only be present if not all constructors were mentioned

If some case is missing, a specific TCC is generated on typechecking

when no ELSE clause has been provided

CASES Expressions
Example

last_inserted_was?(n: string, pb:Phonebook): bool =

CASES pb OF

empty_phonebook: FALSE,

add_phone(name,phone,phonebook): (n = name)

ENDCASES

Implicit Declarations

During typechecking an ADT definition PVS generates:

Definitions for the type, constructors, recognizers, and accessors

As uninterpreted declarations

Additional operators

subterm, <<, reduce_nat, reduce_ordinal

Several axioms

Implicit Declarations
Automatically Generated Axioms

extensionality for constant constructors

- there is only one bottom element for every constant constructor

In the Phonebook example:
Phonebook_empty_phonebook_extensionality: AXIOM

FORALL (e1: (empty?), e2: (empty?)):

e1 = e2;

Implicit Declarations
Automatically Generated Axioms

extensionality for constructors with arguments

- elements are distinguishable by the accessors

In the Phonebook example:
Phonebook_add_phone_extensionality: AXIOM

FORALL (v1: (add_phone?), v2: (add_phone?)):

name(v1) = name(v2) AND

phone(v1) = phone(v2) AND

pb(v1) = pb(v2)

IMPLIES v1 = v2

Implicit Declarations
Automatically Generated Axioms

eta axiom

- using the accessed values to build a new element, results in the same
element

In the Phonebook example:
Phonebook_add_phone_eta: AXIOM

FORALL (v: (add_phone?)):

add_phone(name(v), phone(v), pb(v)) = v

Implicit Declarations
Automatically Generated Axioms

Meaning of accessors

In the Phonebook example:
Phonebook_name_add_phone: AXIOM

FORALL (n: string, p: string, pb: Phonebook):

name(add_phone(n, p, pb)) = n

Phonebook_name_add_phone: AXIOM

FORALL (n: string, p: string, pb: Phonebook):

phone(add_phone(n, p, pb)) = p

Phonebook_name_add_phone: AXIOM

FORALL (n: string, p: string, pb: Phonebook):

phonebook(add_phone(n, p, pb)) = pb

Implicit Declarations
Automatically Generated Axioms

Inclusion and Disjointness

- recognizers characterize all the elements of the type

In the Phonebook example:
Phonebook_inclusive: AXIOM

FORALL (phonebook: Phonebook):

empty?(phonebook) OR add_phone?(phonebook)

Phonebook_disjointness: AXIOM

FORALL (phonebook: Phonebook):

NOT (empty?(phonebook) AND add_phone?(phonebook))

Implicit Declarations
Automatically Generated Axioms

Structural Induction

In the Phonebook example:
Phonebook_induction: AXIOM

FORALL (p: [Phonebook -> boolean]):

(p(empty_phonebook) AND

(FORALL (name: string, phone: string, pb: Phonebook):

p(pb) IMPLIES

p(add_phone(name, phone, pb))))

IMPLIES (FORALL (phonebook: Phonebook): p(phonebook))

Implicit Declarations
Automatically Generated Axioms

subterm(x,y: Phonebook)

- indicates if x participates in the construction of y

In the Phonebook example:
subterm(x: Phonebook, y: Phonebook): boolean =

x = y OR

CASES y

OF empty_phonebook: FALSE,

add_phone(name, phone, phonebook):

subterm(x, phonebook)

ENDCASES;

Implicit Declarations
Automatically Generated Axioms

<<

- relational version of subterm, along with

- an axiom stating its well-foundedness

In the Phonebook example:
<<: (strict_well_founded?[Phonebook]) =

LAMBDA (x, y: Phonebook):

CASES y

OF empty_phonebook: FALSE,

add_phone(name, phone, pb):

x = pb OR x << pb

ENDCASES

Phonebook_well_founded: AXIOM strict_well_founded?[Phonebook](<<)

Implicit Declarations
Automatically Generated Axioms

(Several) Reduce functions

- reduce an element to a natural number, to an ordinal, or to a range.

- useful for simplifying the proof of termination of user-defined functions

In the Phonebook example:
reduce_nat(empty_red: nat,

add_phone_red:[[string,string,nat]->nat])

(phonebook: Phonebook): nat =

LET reduce = reduce_nat(empty_red, add_phone_red)

IN CASES phonebook OF

empty_phonebook: empty?_fun,

add_phone(name, phone, pb): add_phone_red(name, phone, reduce(pb))

ENDCASES

Recursive Definitions

The usual schema for definition by recursion is supported by the
implicit definitions

Example
size(phonebook: Phonebook): RECURSIVE nat =

CASES phonebook OF

empty_phonebook: 0,

add_person(n,p,pb): 1+size(pb)

ENDCASES

MEASURE phonebook BY <<

Formal Parameters

(Incomplete) Declaration Syntax

〈adt name〉
[
[
[
〈arg namei〉: [ TYPE | 〈type〉 ]

]i=1···a
(,)

]
]
: DATATYPE

BEGIN[
〈constructor i〉

[
(
[
〈accessor i,j〉: 〈type〉

]j=1··n
(,)

)
]
: 〈recognizer i〉

]i=1··c

END 〈adt name〉
c, n, a ∈ N

Example: Stack
stack[T: TYPE]: DATATYPE

BEGIN

empty: empty?

push(top:T, pop:stack): nonempty?

END stack

Formal Parameters

For every type parameter combinators every and some are generated

- “a predicate holds for every element included in the structure”
- “a predicate holds for some element included in the structure”

In the Stack example:
every(p: PRED[T])(a: stack): boolean =

CASES a OF

empty: TRUE,

push(push1_var, push2_var):

p(push1_var) AND every(p)(push2_var)

ENDCASES

some(p: PRED[T])(a: stack): boolean =

CASES a OF

empty: FALSE,

push(push1_var, push2_var):

p(push1_var) OR some(p)(push2_var)

ENDCASES

Formal Parameters

A map combinator is also generated

In the Stack example:
map(f: [T -> T1])(a: stack[T]): stack[T1] =

CASES a OF

empty: empty,

push(push1_var, push2_var):

push(f(push1_var), map(f)(push2_var))

ENDCASES

Subtypes

(Incomplete) Declaration Syntax

〈adt name〉
[
[
[
〈arg namei〉: [ TYPE | 〈type〉 ]

]i=1···a
(,)

]
]
: DATATYPE[

WITH SUBTYPES
[
〈subadt namei〉

]i=1··s
(,)

]
BEGIN[

〈constructori〉
[
(
[
〈accessori,j〉: 〈type〉

]j=1··n
(,)

)
]
:〈recognizeri〉

[
:〈subadt namek〉

]]i=1··c

END 〈adt name〉
c, n, a, s ∈ N

Abstract datatypes may also define subtypes

There must be no repetitions in the list of subtype names

Each constructor must return an element of a particular subtype

Implicit declarations are also affected by the introduction of subtypes

Subtypes
Example

ArithExpr: DATATYPE WITH SUBTYPES NumExpr, BoolExpr

BEGIN

CONST (x:real) :constant? : NumExpr

ADD (x1,x2:NumExpr) :addition? : NumExpr

EQUALS(x1,x2:NumExpr) :equals? :BoolExpr

ITE (b:BoolExpr, x1,x2:NumExpr):ite? :NumExpr

END ArithExpr

Some of the implicit declarations

ArithExpr: TYPE

NumExpr(x:ArithExpr): boolean =

constant?(x) OR addition?(x) OR ite?(x)

NumExpr: TYPE =

{x:ArithExpr | constant?(x) OR addition?(x) OR ite?(x)}

BoolExpr(x:ArithExpr): boolean = equals?(x)

BoolExpr: TYPE = (equals?)

Remarks

Datatype definition can be a top-level declaration (as a theory)

If declared in its own file, most of the implicit declarations are
automatically printed out in a read-only separated file
name-of-type_adt.pvs

Also supports IMPORTING and ASSUMING parts

Imported types, parameter types, and subtypes must appear in
positive form

Remarks
Complete Syntax of DATATYPE Declaration

〈adt name〉
[
[
[
〈arg namei〉: [ TYPE | 〈type〉 ]

]i=1···a
(,)

]
]
: DATATYPE[

WITH SUBTYPES
[
〈subadt namei〉

]i=1··s
(,)

]
BEGIN[
IMPORTING

[
〈theory namei〉

]i=1··t
(,)

]
[
BEGIN ASSUMING

[
〈assumptioni〉

]i=1··a′
END ASSUMING

]
[
〈constructor i〉

[
(
[
〈accessor i,j〉: 〈type〉

]j=1··a′′
(,)

)
]
: 〈recognizer i〉

[
:〈subadt namek〉

] ]i=1··c

END 〈adt name〉

Where a, s, t, a′, a′′, c ∈ N

Mariano M. Moscato PVS for Computer Scientists 23 / 31



Enumerated Types

PVS supports an useful abbreviation to define enumerated types

Example

Finger: TYPE = { Thumb, Index, Middle, Ring, Baby}

is an abbreviation for

Finger: DATATYPE

BEGIN

Thumb: Thumb?

Index: Index?

Middle: Middle?

Ring: Ring?

Baby: Baby?

END

Disjoint Unions

Also know as cotuple, coproduct or sum types

Declaration Syntax

〈type name〉: TYPE = [
[
〈type expr〉

]i=1··n
(+)

]

Examples

BoolOrIntOrPred: TYPE = [ bool + int + [int->bool] ]

Either[S,T: TYPE]: TYPE = [ S + T ]

Mariano M. Moscato PVS for Computer Scientists 25 / 31



Disjoint Unions

Also know as cotuple, coproduct or sum types

Declaration Syntax

〈type name〉: TYPE = [
[
〈type expr〉

]i=1··n
(+)

]

For each type in the union, these operators are implicitly generated

Injector IN_i : [〈type expr〉i -> 〈type name〉]
Recognizer IN?_i: [〈type name〉 -> bool]

Extractor OUT_i: [(IN?_i) -> 〈type expr〉i]
in_i, in?_i, and out_i are also generated (with same semantics)

Injectors can be used as discriminators in CASES OF expressions

(!) Union types are not technically abbreviations of ADTs but...

Disjoint Unions
Example

Either[S,T: TYPE]: TYPE = [ S + T ]

duplicate_left_halve_right(data: Either[nat,real]): real =

CASES data OF

IN_1(n): 2*n,

IN_2(x): x/2

ENDCASES

Disjoint Unions
Example

ArithExpr: DATATYPE WITH SUBTYPES NumExpr, BoolExpr

BEGIN

CONST (x:real) :constant? : NumExpr

ADD (x1,x2:NumExpr) :addition? : NumExpr

EQUALS(x1,x2:NumExpr) :equals? :BoolExpr

ITE (b:BoolExpr, x1,x2:NumExpr):ite? :NumExpr

END ArithExpr

Value: TYPE = [ bool + real ]

eval(expr: ArithExpr): RECURSIVE Value =

CASES expr OF

CONST(x): IN_2(x),

ADD(x1,x2): IN_2(OUT_2(eval(x1))+OUT_2(eval(x2))),

EQUALS(x1,x2): IN_2(OUT_2(eval(x1))=OUT_2(eval(x2))),

ITE(b,x1,x2): IF OUT_1(eval(b))

THEN eval(x1)

ELSE eval(x2) ENDIF

ENDCASES

MEASURE expr BY <<

Disjoint Unions
Example

IMPORTING ArithExpr

Value: TYPE = [ bool + real ]

ValueFor(expr: ArithExpr): TYPE =

{v: Value| IF NumExpr(expr) THEN IN?_2(v) ELSE IN?_1(v) ENDIF}

eval(expr: ArithExpr): RECURSIVE ValueFor(expr) =

CASES expr OF

CONST(x): IN_2(x),

ADD(x1,x2): IN_2(OUT_2(eval(x1))+OUT_2(eval(x2))),

EQUALS(x1,x2): IN_2(OUT_2(eval(x1))=OUT_2(eval(x2))),

ITE(b,x1,x2): IF OUT_1(eval(b))

THEN eval(x1)

ELSE eval(x2) ENDIF

ENDCASES

MEASURE expr BY <<

Disjoint Unions
Example

IMPORTING ArithExpr

Value: DATATYPE BEGIN

bool(val:bool):bool?

real(val:real):real?

END Value

ValueFor(expr: ArithExpr): TYPE =

{v: Value| IF NumExpr(expr) THEN real?(v) ELSE bool?(v) ENDIF}

eval(expr: ArithExpr): RECURSIVE ValueFor(expr) =

CASES expr OF

CONST(x): real(x),

ADD(x1,x2): real(val(eval(x1))+val(eval(x2))),

EQUALS(x1,x2): bool(val(eval(x1))=val(eval(x2))),

ITE(b,x1,x2): IF val(eval(b))

THEN eval(x1)

ELSE eval(x2) ENDIF

ENDCASES

MEASURE expr BY <<

